Система Maple. Геометрия на плоскости. Возможности пакета geometry для решения геометрических задач

LOGO

Система Maple. Геометрия на плоскости. Возможности пакета geometry для решения геометрических задач

1. Общая информация о пакете geometry
2. Команды проверки условий для двумерных геометрических объектов
3. Команды определения двумерных геометрических объектов и действий с ними
4. Информационные ресурсы

LOGO

Общая информация о пакете geometry

- Для решения задач планиметрии используется геометрический пакет geometry;
- Перед обращением к его командам сам пакет должен быть загружен при помощи команды with(geometry);
- Для геометрического пакета geometry характерен следующий способ определения объектов: первый параметр команды задаёт имя объекта и далее следует собственно информация об объекте;
- Для просмотра полей структуры, описывающей геометрический объект, используют команду detail(obj);

Общая информация о пакете geometry

- Для графического вывода геометрического объекта используется команда draw(obj);

- При выводе на одном рисунке нескольких геометрических объектов можно указывать параметры графического вывода (например, цвет) каждого объекта;

↓

- Результатом команды являются структуры двумерной или трехмерной графики, и поэтому при обращении к команде draw используются параметры, аналогичные графическим;

LOGO

Команды проверки условий для двумерных геометрических объектов

- По умолчанию _x и _y используются как глобальные переменные для координат точек, а также в качестве переменных в уравнениях прямых и окружностей;

- Геометрические объекты определяются обычным образом: точка задается своими координатами (команда point(name,x1,y1)), прямая – двумя точками или уравнением (команда line), окружность (команда circle) – тремя точками, уравнением, заданием центра и радиуса, диаметром;

- При возможности определенного ответа результатом является булевская константа (true или false); в некоторых случаях выводятся координаты объекта (например, точки), при которых будет выполнено проверяемое условие;

LOGO

Команды проверки условий для двумерных геометрических объектов

- AreCollinear(p1,p2,p3) – проверка условия принадлежности трёх точек p1,p2,p3 одной прямой;

point(name,a,b) – задание точки с координатами a и b

Пример 1. Лежат ли точки на одной прямой? а) A(0,0); B (1,1); C(2,2); б) A(-1,0); B(2,1); C(4,12);
Решение:

```
> AreCollinear(point(A,0,0),point(B,1,1),point(C,2,2));
                                                        true
> AreCollinear(point(A,-1,0),point(B,2,1),point(C,4,12));
                                                        false
```

Ответ: а) лежат на одной прямой; б) не лежат на одной прямой

Команды проверки условий для двумерных геометрических объектов

LOGO

- AreConcurrent(name1,name2,name3) - проверка условия пересечения трёх прямых в одной точке; (где name1,name2,name3 – название линий)

Пример 2. Определить, пересекаются ли в одной точке прямые, заданные уравнениями: $3 \cdot b - 6 = 0$,

$$-\sqrt{3} \cdot a + b + \sqrt{3} - 2 = 0, \quad \sqrt{3} \cdot a + b - \sqrt{3} - 2 = 0$$

Решение:

```
> with(geometry):
> line(1,3*b-6 = 0,[a,b]):
  line(2,-3^(1/2)*a+b+3^(1/2)-2 = 0,[a,b]):
  line(3,3^(1/2)*a+b-3^(1/2)-2 = 0,[a,b]):
AreConcurrent(1, 2, 3);
                                true
```

Ответ: прямые пересекаются в одной точке.

LOGO

Команды проверки условий для двумерных геометрических объектов

- AreConcyclic(p1,p2,p3,p4) - проверка существования окружности, которой принадлежат заданные четыре точки p1,p2,p3,p4

Пример 3. Заданы пять точек своими координатами: p1(0,0); p2(3,0); p3(3,3); p4(0,3); p5(2,8). Проверить, существует ли окружность, которой принадлежат заданные четыре точки: а) p1, p2, p3, p4; б) p1, p2, p3, p5.

Решение:

```
with(geometry):
point(p1,0,0),point(p2,3,0),point(p3,3,3):
point(p4,0,3),point(p5,2,8):
AreConcyclic(p1,p2,p3,p4);
                                                    true
AreConcyclic(p1,p2,p3,p5);
                                                    false
```

Ответ: а) такая окружность существует; б) такой окружности нет.

LOGO

Команды проверки условий для двумерных геометрических объектов

- AreOrthogonal(name1,name2); - проверка условия ортогональности двух геометрических объектов;
- AreParallel(line1,line2); - проверка условия параллельности двух прямых line1,line2;
- Intersection(line1,line,…) – нахождение координаты точки пересечения прямых;

LOGO

Команды проверки условий для двумерных геометрических объектов

Пример 4. Проверить, являются ли ортогональными окружности, а) c1 и c2; б) c2 и c3, заданные соответствующими уравнениями:

$c1 : x^2 + y^2 = 1,\ \ c2 : (x-2)^2 + y\hat{}2 = 2,\quad c3 : x^2 + y^2 = 2$

Решение:

```
> with(geometry):
> _EnvHorizontalName := 'x': _EnvVerticalName := 'y':
  circle(c1,x^2 + y^2 =1), circle(c2, (x-2)^2 + y^2 = 2):
  circle(c3, x^2 + y^2 = 2):
  AreOrthogonal(c1,c2);
                                                        false
> AreOrthogonal(c2,c3);
                                                        true
```

Ответ: а) не являются ортогональными; б) ортогональны

LOGO

Команды проверки условий для двумерных геометрических объектов

Пример 5. Проверить условие параллельности прямых а) АВ и AF; б)АВ и CD, если A(0,1), B(1,0), F(1,1), CD: x+y=2.

Решение:

```
> with(geometry):
> point(A,[0,1]);
  point(B,[1,0]);
  point(F,[1,1]);
                                        A
                                        B
                                        F
> line(AB,[A,B]):
  line(AF,[A,F]);
                                        AB
                                        AF
> line(CD,x+y=2,[x,y]);
                                        CD
> AreParallel(AB,AF);
                                        false
> AreParallel(AB,CD);
                                        true
```

Ответ: а) АВ не параллельна AF, б) АВ параллельна CD.

Команды проверки условий для двумерных геометрических объектов

- ArePerpendicular(line1,line2) – проверка условия перпендикулярности двух прямых line1 и line2

Пример 6. Пусть три прямы заданы соответствующими уравнениями: l1: y=x, l2: y=-x, l3: x=2. Выяснить, являются ли перпендикулярными прямые: а) l1 и l2; б) l1 и l3.

Решение:

```
> with(geometry):
> line(l1,y=x,[x,y]):
  line(l2,y=-x,[x,y]):
  ArePerpendicular(l1,l2);
                                                    true
> line(l3,x=2,[x,y]):
  ArePerpendicular(l1,l3);
                                                    false
> with(geometry):
```

Ответ: а) l1 перпендикулярна l2; б) l1 не перпендикулярна l3.

LOGO

Команды проверки условий для двумерных геометрических объектов

- AreSimilar(T1,T2); - проверка условия подобия двух треугольников T1 и T2
- triangle(name,[p1,p2,p3]) – задание треугольника тремя точками p1,p2,p3, тремя прямыми или тремя сторонами

Пример 7. Пусть заданы точки своими координатами: A(0,0); B (1,3); C(1,0); H(0,6); F(2,0). Подобен ли треугольник ABC треугольнику AHF?

Решение:

```
> with(geometry):
> point(A,0,0),point(B,1,3),point(C,1,0),point(H,0,6),point(F,2,0):
  triangle(T1, [A, B, C]):
  triangle(T2, [A, H, F]):
  AreSimilar(T1, T2);
                                                                true
```

Ответ: треугольник ABC подобен треугольнику AHF.

Команды проверки условий для двумерных геометрических объектов

LOGO

- AreTangent(NAME_line,NAME_circle) – проверка, является ли прямая line касательной окружности circle

Пример 8. Пусть прямая задана уравнением $2x+3y=0$, и даны две окружности, также заданные соответствующими уравнениями: $(x-2)^2+y^2=1$ $x^2+y^2=1$, Выяснить, к какой окружности прямая является касательной.

Решение:

```
> with(geometry):
> _EnvHorizontalName := 'x': _EnvVerticalName := 'y':
  circle(c1,x^2 + y^2 =1), circle(c2,(x-2)^2 + y^2 =1):
  line(l, 2*x + 3*y =0):
  AreTangent(c1, c2);
                                                          true
> AreTangent(l, c1);
                                                          false
```

$x^2+y^2=1,$

LOGO

Команды проверки условий для двумерных геометрических объектов

- IsEquilateral(name) – проверка треугольника name на равносторонность

Пример 9. Пусть вершины треугольника ABC заданы своими координатами: A(0,0); B(2,0); C(1,2). Проверить, является ли треугольник ABC равносторонним.

Решение:

```
> with(geometry):
> triangle(ABC, [point(A,0,0), point(B,2,0), point(C,1,2)]);
                                                              ABC
> IsEquilateral(ABC);
                                                              false
```

Ответ: треугольник ABC не является равносторонним.

LOGO

Команды проверки условий для двумерных геометрических объектов

- IsRightTriangle(name) – проверка, является ли треугольник name прямоугольным

Пример 10. Пусть вершины треугольника ABC заданы своими координатами: A(0,0); B(2,0); C(0,2). Выяснить, является ли треугольник ABC прямоугольным.

Решение:

```
> with(geometry):
> triangle(ABC, [point(A,0,0), point(B,2,0), point(C,0,2)]);
                                                            ABC
> IsRightTriangle(ABC);
                                                            true
```

Ответ: треугольник ABC прямоугольный.

Команды проверки условий для двумерных геометрических объектов

- IsOnCircle(pt,circle) – проверка условия принадлежности точки pt окружности circle
- IsOnLine(pt,line) - проверка условия принадлежности точки pt прямой line

Пример 11. Принадлежит ли точка A(-1,0) окружности, заданной уравнением $x^2 + y^2 = 1$?

Решение:

```
> with(geometry):
:
> circle(c1,x^2 + y^2 =1,[x,y]), point(A,-1,0):
  IsOnCircle(A, c1);
                                                        true
.
```

Ответ: точка A(-1,0) принадлежит окружности.

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- area(name) – вычисление площади заданного объекта name(треугольника, круга или квадрата)

Пример 12. Найти площадь треугольника ABC, заданного координатами своих вершин: A(0,0); B(2,0); C(1,3).

Решение:

```
> with(geometry):
> triangle(ABC, [point(A,0,0), point(B,2,0), point(C,1,3)]):
  area(ABC);
                                        3
```

Ответ 3

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- center (name, circle) - позволяет определить центр окружности, результат присваивается переменной name; coordinates(pt) - вывод координат точки pt

Пример 13. Найти координаты центра окружности, заданной уравнением: $(x-3)^2+(y-1)^2=4$

Решение:

```
> with(geometry):
> _EnvHorizontalName := x: _EnvVerticalName := y:
  circle(c, (x-3)^2 + (y-1)^2 =4);
                                                        c
> center(c);
                                                  center_c
> coordinates(center(o,c));
                                                    [3, 1]
```

Ответ: (3,1).

Команды определения двумерных геометрических объектов и действий с ними

- centroid(name,tri) – команда, позволяющая вычислить центр тяжести треугольника.

Пример 14. Вычислить координаты центра тяжести треугольника ABC, заданного координатами своих вершин: A (0,0); B(2,0); C(1,3).

Решение:

```
> with(geometry):
> triangle(ABC, [point(A,0,0), point(B,2,0), point(C,1,3)]):
> centroid(g,ABC);
                                                    g
> coordinates(g);
                                                  [1, 1]
```

Ответ: (1,1).

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- Circumcircle(name,tri) – вычисление описанной вокруг треугольника tri окружности;
- diagonal(Sq) – вычисление длины диагонали квадрата Sq;
- diameter([pt1,pt2,…]) вычисление диаметра круга, содержащего заданные точки;
- incircle(name,tri) – вычисление вписанной в треугольник tri окружности;
- line(p1,p2) – определение прямой, заданной двумя точками или уравнением;

Команды определения двумерных геометрических объектов и действий с ними

Пример 15. Вычислить и изобразить графически описанную окружность вокруг треугольника ABC, заданного координатами своих вершин: A(0,0); B(2,0); C(1,3).

Решение:

```
> with(geometry):
> triangle(T, [point(A,0,0), point(B,2,0), point(C,1,3)]):
> circumcircle(Elc, T, 'centername' = OO);
```

Elc

```
> detail(Elc);
```

*name of the object: Elcform of the object: circle2dname of the center: OOcoordinates of the center: [1, 4/3]radius of the circle: 1/9*25^(1/2)*9^(1/2)equation of the circle:*

*x^2+y^2-2*x-8/3*y = 0*

```
> draw({Elc,T},printtext=true);
```

LOGO

Команды определения двумерных геометрических объектов и действий с ними

```
draw({Elc,T},printtext=true);
```



Команды определения двумерных геометрических объектов и действий с ними

Пример 16. Найти длину диагонали квадрата ABCF, заданного координатами своих вершин: A(0,0); B(1,0); C(1,1); F(0,1).

Решение:

```
> with(geometry):
> point(A,0,0),point(B,1,0),point(C,1,1),point(F,0,1):
> square(Sq,[A,B,C,F]);
```

$$Sq$$

```
> diagonal(Sq);
```

$$\sqrt{2}$$

Ответ:

$$\sqrt{2}$$

LOGO

Команды определения двумерных геометрических объектов и действий с ними

Пример 17. Вычислить диаметр круга, содержащего точки A (0,0); B(2,0); C(1,3); F(1,6); M(2^(1/2),3)

Решение:

```
> with(geometry):
:
> (A,0,0),point(B,2,0),point(C,1,3),point(F,1,6):
  point(M,sqrt(2),3):
  ps := [A, B, C, F, M]:
  diameter(ps);
```

$$[A, F, \sqrt{37}]$$

```
:
>
```

Ответ: $\sqrt{37}$

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- distance(pt,line) – нахождение расстояния между точкой pt и прямой line. В качестве второго параметра может фигурировать точка, тогда вычисляется расстояние между двумя точками.

Пример 18. Найти расстояние между точками A и B, заданными своими координатами: A(a,b); B(c,d).

Решение:

```
> with(geometry):
> point(A,a,b), point(B,c,d):
  distance(A,B);
```

$$\sqrt{(a-c)^2+(b-d)^2}$$

Ответ: $\sqrt{(a-c)^2+(b-d)^2}$

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- ellipse(name,uravnenie) – определение эллипса одним из следующих способов: по пяти точкам, по центру и двум полуосям или при помощи уравнения.

Пример 19. Найти координаты центра эллипса, заданного уравнением: $2x^2 + y^2 - 4x + 4y = 0$

Решение:

```
> with(geometry):
> _EnvHorizontalName := 'x': _EnvVerticalName := 'y':
> ellipse(e1,2*x^2+y^2-4*x+4*y=0);
                                                        e1
> center(e1), coordinates(center(e1));
                                              center_e1, [1, -2]
```

Ответ: (1,-2).

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- FindAngle(l1,l2) – вычисление угла между двумя прямыми l1 и l2 или двумя окружностями

Пример 20. Найдите угол между двумя прямыми, заданными соответствующими уравнениями: x+y=1 и x-y=1

Решение:

```
> with(geometry):
> _EnvHorizontalName := 'x': _EnvVerticalName := 'y':
> line(l1,x + y = 1), line(l2,x - y =1);
```

$$l1, l2$$

```
> FindAngle(l1, l2);
```

$$\frac{1}{2}\pi$$

Ответ: $\frac{1}{2}\pi$

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- Hyperbola – определение гиперболы, задаваемой набором точек или другими характеристиками;

Пример 21: Найти координаты центра гиперболы, заданной уравнением $9y^2 - 4x^2 = 36$

Решение:

```
> with(geometry):
> hyperbola(h1,9*y^2-4*x^2=36,[x,y]):
  center(h1), coordinates(center(h1));
```

$$center\_h1, [0, 0]$$

Ответ: (0,0).

LOGO

Команды определения двумерных геометрических объектов и действий с ними

Пример 22: Вычислить и изобразить графически вписанную окружность в треугольник ABC, заданного координатами своих вершин: A(0,0); B(2,0); C(1,3).

Решение:

```
> with(geometry):
> triangle(T, [point(A,0,0), point(B,2,0), point(C,1,3)]):
> incircle(inc,T,'centername'=o);
                                   inc
> detail(inc);
name of the object: incform of the object: circle2dname of the center: ocoordinates of the center: [1, 3/(10^(1/2)+1)]radius of the circle: 3/(10^(1/2)+1)equation of the circl
    1+x^2+y^2-2*x-6/(10^(1/2)+1)*y = 0
```

LOGO

Команды определения двумерных геометрических объектов и действий с ними

```
> draw({inc,T},printtext=true);
```



LOGO

Команды определения двумерных геометрических объектов и действий с ними

- intersection(pt,obj1.obg2) – вычисление точки пересечения двух прямых или двух окружностей;

Пример 23: Найти координаты точки пересечения двух прямых, заданных уравнениями: x=0 и x+y=1.

Решение:

```
> with(geometry):
> line(l1, x = 0, [x,y]), line(l2, x + y = 1, [x,y]):
> intersection(G, l1, l2);
                                                    G
> coordinates(G);
                                                    [0, 1]
```

Ответ: (0,1).

Команды определения двумерных геометрических объектов и действий с ними

LOGO

- median(name,A,tri) – определение медианы треугольника tri, проведенной из вершины A

Пример 24: Найти уравнение медианы, проведенной из вершины A, треугольника ABC, заданного координатами своих вершин: A(0,0); B(2,0); C(1,3).

Решение:

```
> with(geometry):
> triangle(ABC, [point(A,0,0), point(B,2,0), point(C,1,3)]):
> median(mA, A, ABC);
```

mA

```
> detail(mA);
```

name of the object: mAform of the object: line2dequation of the line: -3/2*x+3/2*y = 0

Ответ: -3/2*x+3/2*y=0.

Команды определения двумерных геометрических объектов и действий с ними

- midpoint(name,pt1,pt2) – вычисление средней точки на отрезке, заданном двумя точками pt1 и pt2

Пример 25: Найти координаты середины отрезка AB, если A (0,0); B(2,0).

Решение:

```
> with(geometry):
> point(A,0,0), point(B,2,0):
  midpoint(C1,A, B);
```

$$C1$$

```
> coordinates(C1);
```

$$[1, 0]$$

Ответ: (1,0).

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- parabola(name,…) – задание параболы набором точек или другими характеристиками;
- focus(obj) – определение фокуса объекта obj;

Пример 26: Задайте параболу с помощью уравнения $y^2+12x-6y+33=0$. Найдите координаты фокуса параболы.

Решение:

```
> with(geometry):
> parabola(p1,y^2+12*x-6*y+33=0,[x,y]);
```

$$p1$$

```
> focus(p1), coordinates(focus(p1));
```

$$focus\_p1, [-5, 3]$$

Ответ: (-5,3).

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- ParallelLine(name,pt,line) – вычисление прямой, проходящей через точку pt и параллельной прямой line

Пример 27: Найдите уравнение прямой, проходящей через точку P(2,3) и параллельной прямой x+y=1.

Решение:

```
> with(geometry):
> point(P, 2 , 3), line(l,x + y =1,[x,y]);
```

$$P, l$$

```
> ParallelLine(lp, P, l);
```

$$lp$$

```
> detail(lp);
```

name of the object: lpform of the object: line2dequation of the line: $-5+x+y=0$

Ответ: -5+x+y=0

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- PerpenBisector(name,pt1,pt2) – вычисление прямой, проходящей через середину отрезка, заданного двумя точками pt1 и pt2, и ортогональной ему;

Пример 28: Найти уравнение прямой, проходящей через середину отрезка, заданного двумя точками A(0,0); B(2,0), и ортогональной ему.

Решение:

```
> with(geometry):
> point(A,0,0), point(B,2,0);
                                        A, B
> PerpenBisector(l, A, B);
                                        l
> detail(l);
        name of the object: lform of the object: line2dequation of the line: -2+2*x = 0
```

Ответ: -2+2x=0.

LOGO

Команды определения двумерных геометрических объектов и действий с ними

❖ PerpendicularLine(name,pt,line) – вычисление прямой, проходящей через точку pt и перпендикулярной прямой line

Пример 29: Найти уравнение прямой, проходящей через точку P(2,3) и перпендикулярной прямой, заданной уравнением x+y=1.

Решение:

```
> with(geometry):
> point(P, 2 , 3), line(l, x + y =1, [x,y]);
                                                P,l
> PerpendicularLine(lp, P, l);
                                                lp
> detail(lp);
          name of the object: lpform of the object: line2dequation of the line: 1+x-y = 0
```

Ответ: 1+x-y=0.

Команды определения двумерных геометрических объектов и действий с ними

- radius(circle) – вычисление радиуса окружности circle

Пример 30: Найти радиус окружности, заданной уравнением

$$x^2 + y^2 = 9$$

Решение:

```
> with(geometry):
> circle(c, x^2 + y^2 = 9,[x,y]);
```

c

```
> radius(c);
```

$\sqrt{9}$

Ответ: 3

Команды определения двумерных геометрических объектов и действий с ними

- sides(obj) – вычисление периметра треугольника или квадрата
- square(name,[pt1,pt2,pt3,pt4]) – задание квадрата четырьмя точками

Пример 31. Вычислить периметр квадрата ABEC, заданного координатами своих вершин: A(0,0); B(3,0); E(3,3); C(0,3).

Решение:

```
with(geometry):
point(A,0,0), point(B,3,0), point(C,0,3),point(E,3,3):

square(Sq,[A,B,E,C]):
sides(Sq);
```

$$\frac{1}{2}\sqrt{18}\,\sqrt{2}$$

Ответ: $\frac{1}{2}\sqrt{18}\sqrt{2}$

LOGO

Команды определения двумерных геометрических объектов и действий с ними

- TangentLine(name,pt,circle) – вычисление двух прямых, проходящих через точку pt и касательных к окружности circle; результат присваивается переменной name

Пример 32. Найти уравнения прямых, проходящих через точку A(1,1) и являющихся касательными к окружности $a^2+b^2=1$

Решение:

```
> with(geometry):
> point(A, 1, 1), circle(c, a^2 + b^2 = 1,[a,b]);
```

$$A, c$$

```
> TangentLine(obj, A, c, [l1, l2]);
```

$$[l1, l2]$$

```
> form(l1), Equation(l1);
```

$$line2d, a-1=0$$

```
> form(l2), Equation(l2);
```

$$line2d, 1-b=0$$

Ответ: a-1=0; 1-b=0

Команды определения двумерных геометрических объектов и действий с ними

- Tangentpc(name,pt,circle) – вычисление касательной к окружности circle, проходящей через точку pt

Пример 33: Найти уравнение касательной к окружности

$$x^2 + y^2 = 1$$

проходящей через точку A(1,0).

Решение:

```
> with(geometry):
> point(A, 1, 0), circle(c, x^2 + y^2 = 1, [x,y]);
                                                        A, c
> tangentpc(l, A, c);
                                                        l
> Equation(l);
                                                        x - 1 = 0
```

Ответ: x-1=0.

LOGO

Информационные ресурсы

1) Говорухин В., Цибулин В. Компьютер в математическом исследовании. Учебный курс. СПб.: Питер, 2001.-624 с.

2) Иллюстрированный самоучитель по Maple: http://www.knigka.info/2007/10/30/illjustrirovannyjj_samouchitel_po_maple.html

3) Электронный учебник по Maple 7: http://math-guru.ru/